



普通高等教育“十二五”规划教材

程序设计 及数据库编程 实践教程

主编 张新明 沈俊媛
何 锋 胡 丹



本书附音像光盘



科学出版社

普通高等教育“十二五”规划教材

程序设计及数据库编程实践教程

主 编 张新明 沈俊媛 何 锋 胡 丹

科学出版社

北 京

内 容 简 介

本书与《程序设计及数据库编程教程》配套使用,主要包括: Visual Basic 基础知识实验、窗体和常用控件实验、程序设计基础实验、程序控制结构实验、数组实验、过程和函数实验、创建 Access 数据库实验、SQL 语句实验、数据库的安全实验、Visual Basic 数据库编程实验、数据文件访问及菜单界面设计实验。还设计制作了三个数据库应用系统开发案例:小区收费管理、学生成绩管理、网上书店管理。

光盘内容:实验源文件、实验素材、操作演示视频、教学 PPT。

本书可作为高校非计算机专业的 VB 程序设计+数据库应用基础课程的实验教材,也可供计算机爱好者参考使用。

图书在版编目(CIP)数据

程序设计及数据库编程教程:含实践教程/陈丽花等主编. --北京:科学出版社,2014

普通高等教育“十二五”规划教材

ISBN 978-7-03-039660-0

I. ①程… II. ①陈… III. ①BASIC 语言-程序设计-高等学校-教材
②数据库管理系统-程序设计-高等学校-教材 IV. ①TP312
②TP311.13

中国版本图书馆 CIP 数据核字(2014)第 017777 号

责任编辑:李淑丽 / 责任校对:刘亚琦

责任印制:肖 兴 / 封面设计:华路天然工作室

科学出版社 出版

北京东黄城根北街 16 号

邮政编码:100717

<http://www.sciencep.com>

新科印刷有限公司 印刷

科学出版社发行 各地新华书店经销

2014 年 2 月第 一 版 开本:787×1092 1/16

2014 年 2 月第一次印刷 印张:8

字数:189 000

定价:56.00 元(全套含光盘)

(如有印装质量问题,我社负责调换)

前 言

把程序设计与数据库应用两门课程整合进行联动改革，将程序设计与数据库知识贯通，培养非计算机专业学生程序设计与数据库编程的思维和能力的提高，提高其应用和开发信息管理系统的能力和水平，是我们编写教材希望达到的教学目标。本书是与《程序设计及数据库编程教程》教材配套使用的上机实验教程。根据“培养计算思维能力”的主导思想，根据非计算机专业学生的学习需求，总结多年来独立进行程序设计与数据库课程教学的经验得失，制定教学目标，组织实验内容编写而成，满足一般院校计算机基础课程的教学需要。

本书包括 11 章，由浅入深，由简单到复杂，精心组织编写了 34 个基本实验或综合实验，还有三个数据库管理应用系统的开发案例。为了便于学生独立完成实验，在实验内容及操作指导中给出了相应的操作说明或提示，特别在随书光盘中，有理论教程各章习题及答案，编写本书各章的老师都制作了教学 PPT，给出了实验所需文件，有的还制作了操作演示视频和实验的结果文件，易学易教。特别是：三个数据库管理应用系统开发案例，不但提供了代码、结果文件（.EXE），还提供了开发制作过程的演示视频，为学生自行设计制作出优秀的数据库应用系统作品，给以启示。

本书作者均为一线教师，有丰富的教学经验，沙莉编写第 1 章，陈丽花编写第 2 章，李莉平编写第 3 章，何锋编写第 4 章，胡丹编写第 5 章，沈湘芸编写第 6 章，徐娟编写第 7 章，沈俊媛编写第 8 章，李春宏编写第 9 章和综合案例二，玄文启编写第 10 章，张新明编写第 11 章和综合案例一，陶冶编写综合案例三。全书由张新明统稿。

本书的编写得到了云南财经大学各级领导的关心和支持，在此表示深深地感谢！此外，还要感谢科学出版社的各级领导和相关工作人员对我们编写教材的精心策划、组织和编辑！没有他们的帮助和支持就没有本书的问世。

由于时间仓促，编者水平有限，书中难免有不妥之处，敬请各位读者批评指正！

编 者

2013 年 10 月

目 录

第 1 章 Visual Basic 基础知识实验	1
实验 1.1 熟悉 Visual Basic 6.0 集成开发环境	1
实验 1.2 设计一个 VB 应用程序	2
第 2 章 Visual Basic 窗体和常用控件实验	5
实验 2.1 基本控件程序设计	5
实验 2.2 综合实验	15
第 3 章 Visual Basic 程序设计基础实验	19
实验 3.1 计算年龄	19
实验 3.2 算术运算	20
实验 3.3 交换数	21
实验 3.4 转换时间格式	22
第 4 章 程序控制结构实验	24
实验 4.1 编写一个可以计算个人应交纳所得税的程序	24
实验 4.2 编写一个演示字体及其效果的控制程序	25
实验 4.3 设计一个管理系统的用户登陆身份验证程序	27
实验 4.4 设计一个管理系统的用户管理程序	29
实验 4.5 设计一个管理系统的编辑用户信息程序	32
第 5 章 数组实验	35
第 6 章 过程和函数实验	39
实验 6.1 验证哥德巴赫猜想	39
实验 6.2 输入三个数, 求出它们中的最大数	40
实验 6.3 数组元素的和	41
实验 6.4 求任意两个整数的最大公约数	42
第 7 章 创建 Access 数据库实验	44
实验 7.1 Access 开发环境	44
实验 7.2 创建数据库	46
实验 7.3 创建数据表	46
实验 7.4 设置数据表的字段属性	49
实验 7.5 创建数据表 (二)	50

实验 7.6 建立表间的关系	53
第 8 章 SQL 语句实验	55
实验 8.1 DQL 数据查询语言	55
实验 8.2 DML 数据操纵语言	61
实验 8.3 DDL 数据定义语言	68
第 9 章 数据库的安全实验	73
实验 9.1 数据库编码/解码	73
实验 9.2 生成 MDE 文件	73
第 10 章 Visual Basic 数据库编程实验	75
实验 10.1 使用 Data 控件访问数据库	75
实验 10.2 使用 ADO 控件访问数据库	75
实验 10.3 API 编程	76
实验 10.4 基于 VB 的网络编程	76
实验 10.5 基于 VB 的图形图像处理与动画设计方法	77
实验 10.6 基于 VB 的文件系统控件与编程	79
实验 10.7 VB 工程打包方法与实现	80
第 11 章 数据文件访问及菜单界面设计实验	82
实验 11.1 创建一个单窗体界面的简易记事本	82
实验 11.2 创建多文档界面的应用程序	85
综合案例一 小区物业收费系统开发	89
综合案例二 学生成绩管理系统开发	99
综合案例三 网上书店管理系统开发	110
参考文献	122

第 1 章 Visual Basic 基础知识实验

实验 1.1 熟悉 Visual Basic 6.0 集成开发环境

1. 实验目的

- (1) 了解 Visual Basic 6.0 集成开发环境。
- (2) 掌握窗体中控件对象的建立、选定、调整布局、属性设置和删除等操作方法。
- (3) 掌握 Visual Basic 6.0 帮助系统的使用。

2. 实验内容

- (1) Visual Basic 6.0 集成环境的基本使用：启动和关闭 Visual Basic 6.0 系统；属性窗口、工程资源管理器窗口、代码窗口、窗体布局窗口、立即窗口、工具箱的显示与关闭。
- (2) 控件的基本操作：将工具箱中的每一个常用控件添加到窗体中，并识别这些控件，调整其位置、大小；修改窗体、标签、命令按钮的 Caption、Font 等属性，观察它们的变化。
- (3) Visual Basic 6.0 帮助系统的使用。

3. 实验操作

- (1) 单击 Visual Basic 6.0 集成开发环境的“帮助”菜单，弹出其下拉菜单，选择下拉菜单中的“内容”或“索引”命令，可以打开“MSDN Library Visual Studio 6.0”窗口界面，如图 1-1 所示。分别通过“目录”、“索引”、“搜索”来查找需要帮助的信息。



图 1-1 MSDN Library Visual Studio 6.0 窗口

- (2) 在 Visual Basic 界面的任何上下文相关部分按 F1 键，就可显示有关该部分的信息，上下文相关部分包括：Visual Basic 中的每个窗口（“属性”窗口、“代码”窗口等）、工具箱中的控件、窗体或文档对象内的对象、“属性”窗口中的属性、Visual Basic 关键词（语句、声明、函数、属性、方法、事件和特殊对象）、错误信息。

(3) 从 Internet 上获得帮助: Visual Basic 主页, 该站点包含: Visual Basic 基础知识、Visual Basic 软件库、Visual Basic 常见问题等, 地址为 <http://www.microsoft.com/vbasic/>; 还有一些 Visual Basic 学习的网站:

学 VB.NET	http://www.xuevb.net/
VB 爱好者乐园	http://www.vbgood.com/vb.good/
问专家	http://www.china-askpro.com/
中国 VB 网	http://www.chinavb.net/
VB 大世界	http://vbworld.sxw.gov.cn/vbbooks/index.asp

实验 1.2 设计一个 VB 应用程序

1. 实验目的

- (1) 理解 VB 中对象以及对象的属性、事件和方法的概念。
- (2) 熟悉 VB 编程的基本步骤: 界面设计、属性设计、编写代码、运行、调试、保存文件。
- (3) 掌握在属性窗口中设置对象属性的基本操作。
- (4) 掌握在代码窗口中编辑程序代码的基本操作。
- (5) 体验 Visual Basic 程序编写过程, 达到对可视化编程有一定感性认识的目的。

2. 实验内容

编写一个程序。程序的功能: 在窗体中单击按钮, 显示 “Visual Basic 6.0 欢迎您!”, 再次单击按钮, 则隐藏字符, 同时按钮上的标识在 “显示” 和 “隐藏” 之间切换。

3. 实验操作

大家可能会对操作过程中的操作和代码不了解, 此时我们不需要去探究, 在以后的章节会逐步详细的介绍 Visual Basic 的编程方法和技巧, 在此处用户只需要根据操作步骤来完成本实验。

(1) 启动 Visual Basic 6.0 程序, 系统新建一个标准 EXE 工程, 即工程中包含一个默认的窗体 Form1, 如图 1-2 所示。




图 1-2 显示窗体

(2) 进行界面设计：设计好的窗口界面如图 1-3 所示。界面应该包含_____、_____控件，“显示”应该设置_____控件的_____属性，它的字号应该用_____属性设置，其他隐含的属性有_____。




图 1-3 已添加控件的窗口界面

可以参考表 1-1 在属性框中进行属性设置。

表 1-1 在属性框中设置的各控件属性

对象名	属性名	默认值	设置值
Form1	Caption	Form1	Visual Basic 欢迎您！
Label1	Caption	Label1	空
	Alignment	0-Left Justify	2-Center
	Font	宋体、常规、小五号	隶书、粗体、一号
Command1	Caption	Command1	显示
	Font	宋体、常规、小五号	宋体、粗体、小四

(3) 根据题意“在窗体中单击按钮”，单击指对象的三要素中的_____，按钮是指_____，“再次单击按钮，则隐藏字符”应该用方法实现。

显示“Visual Basic 6.0 欢迎您!”应该用_____控件的_____属性设置，显示“显示”或“隐藏”字样应该用_____控件的_____属性设置，在代码窗口中编写相应的程序代码：

```
Private sub command1_click()                                'command1 的 Click 事件过程
If label1.caption= "" then
    Label1.caption= "Visual Basic 6.0 欢迎您!" '标签显示"Visual Basic 6.0 欢迎您!"
    Command1.caption= "隐藏"                '命令按钮显示"隐藏"字样
Else
    Label1.caption= ""                        '标签显示为空
    Command1.caption= "显示"                '命令按钮显示"显示"字样
End if
End sub
```

(4) 在工具栏上单击“启动”按钮，以查看程序运行结果：屏幕上出现“Visual Basic 6.0 欢迎您!”窗体，窗体上有一“显示”按钮，如图 1-4 所示，单击该按钮，按钮上方出

现“Visual Basic 6.0 欢迎您!”字样,如图 1-5 所示,于此同时,按钮名由“显示”变为“隐藏”,如果再次单击“隐藏”按钮,则窗体显示为如图 1-3 所示,即循环显示。



图 1-4 运行窗体



图 1-5 显示欢迎字样




图 1-6 生成可执行文件

(5) 将此工程生成可执行文件,可执行文件是以 exe 为后缀的文件,不需要在 Visual Basic 环境下即可运行的文件。从菜单栏中选择“文件”→“生成工程 1.exe”命令,如图 1-6 所示。

(6) 选择“文件”→“保存工程”命令,分别保存窗体文件(图 1-7)和工程文件(图 1-8)在指定的文件夹中。

(7) 提示:①单引号用来标记注释信息,一个程序行上单引号之后的所有代码均为注释信息;②双引号用来标记字符串,括在一对双引号之间的内容均为字符,二

者的作用完全不相同,一定不要混淆;③程序中使用的标点符号都应该是西文输入方式下输入的,如果使用了中文标点符号,也会导致语法错误;④程序录入出现的语法错误:系统用红色表示一个语句行,表示该语句行有语法错误。这类错误通常是关键字拼写错误、使用了中文的标点符号、括号不匹配、遗漏了某些标点符号、函数调用缺少参数、不按格式书写语句定义符等;另外,正常情况下,当一行语句输入完按下回车键后,代码中对象名的首字母将变成大写,若没有转换成大写字母,就表示设计的用户界面中没有创建该控件,在运行程序时会出现运行错误。




图 1-7 保存窗体文件




图 1-8 保存工程窗口

第 2 章 Visual Basic 窗体和常用控件实验

实验 2.1 基本控件程序设计

1. 实验目的

- (1) 掌握标签、文本框、命令按钮控件的常用属性、事件和方法。
- (2) 掌握滚动条控件、时钟控件的常用属性、重要事件和基本方法。
- (3) 掌握图片框、图像框和直线等控件的常用属性、重要事件和基本方法。
- (4) 掌握单选按钮、复选框的常用属性、重要事件和基本方法。
- (5) 掌握列表框、组合框的常用属性、重要事件和基本方法。
- (6) 熟练掌握在窗体上建立上述控件的操作方法，以及编写事件过程的程序设计。

2. 实验内容

(1) 显示或清除图片。窗体上有三个按钮“显示图形 (L)”、“清除图形 (C)”、“结束 (E)”和一个图像框，单击“显示图形 (L)”按钮则在图像框中显示一个图片，单击“清除图形 (C)”按钮则清除图像框中的图片，单击“结束 (E)”按钮则结束程序的运行。如图 2-1 所示。

(2) 模拟秒表程序。单击“开始”按钮，该按钮变为灰色，同时“停止”按钮可用，在文本框 1 中显示开始时间，单击“停止”按钮，该按钮变为灰色，同时“开始”按钮可用，在文本框 2 中显示结束时间，在文本框 3 中显示共用时间。

(3) 图像框、滚动条、计时器实验。编写一个动画程序，单击“开始”按钮，图像从左到右不断运动，然后“开始”按钮变成不可用，同时“停止”按钮变成可用，用滚动条来控制运动的速度。

(4) 简单文本设置。编写一个设置文本样式和字号的应用程序，其主要功能是在文本框中输入文本，并且可以设置文本的样式和字号。

(5) 左右移动字幕。为一个应用系统设计滚动字幕版，字幕“天长地久有时尽，此恨绵绵无绝期”在窗体中从左至右后从右向左反复移动，移动速度和移动幅度可以由用户控制。

(6) 列表框应用举例。设计一个窗体，窗体中有 2 个列表框 (List1、List2) 和 8 个按钮。此程序能实现两个列表框之间的信息互换，且 List1 中只允许单选，List2 中允许多选且信息可以排序。

3. 实验操作

1) 显示或清除图片

- (1) 在窗体上添加一个图像框和三个命令按钮。属性设置如表 2-1 所示。

表 2-1 控件属性设置

对象类型	对象名称	属性名	属性值
命令按钮	Command1	Caption	显示图形 (&L)

续表

对象类型	对象名称	属性名	属性值
命令按钮	Command2	Caption	清除图形 (C)
命令按钮	Command3	Caption	结束 (E)
图像框	Image1	Stretch	true

(2) 程序代码如下：

```
Private sub command1_click()  
    Image1.picture= loadpicture("e:\VB\aa.bmp")  
End sub  
Private sub command2_click()  
    Image1.picture= loadpicture("")  
End sub  
Private sub command3_click()  
    End  
End sub
```

说明：Image1.picture = loadpicture (“e: \ VB \ aa.bmp”) 中的 “e: \ VB \ aa.bmp” 是图片文件的文件名，具体操作的时候选择用户磁盘上的图片文件。运行结果如图 2-1 所示。

2) 模拟秒表程序

(1) 在窗体上添加 2 个命令按钮、3 个标签、3 个文本框，如图 2-2 所示。属性设置如表 2-2。




图 2-1 运行界面




图 2-2 窗体设置界面

表 2-2 控件属性设置

对象类型	对象名称	属性名	属性值
命令按钮	Command1	Caption	开始 (S)
命令按钮	Command2	Caption	停止 (S)
标签	Label1	Caption	开始
	Label2	Caption	结束
	Label3	Caption	共用时
文本框	Text1	Text	“ ”
	Text2	Text	“ ”
	Text3	Text	“ ”

(2) 程序代码如下:

```
Dim st, et, pt
Private Sub Command1_Click()
    st = Now
    Text1 = Format(st, "hh:mm:ss")
    Text2 = ""
    Text3 = ""
    Command1.Enabled = False
    Command2.Enabled = True
End Sub
Private Sub Command2_Click()
    et = Now
    pt = et - st
    Text2 = Format(et, "hh:mm:ss")
    Text3 = Format(pt, "hh:mm:ss")
    Command2.Enabled = False
    Command1.Enabled = True
End Sub
```



图 2-3 运行界面

运行结果如图 2-3 所示。

3) 图像框、滚动条、计时器实验

(1) 在窗体上添加 2 个命令按钮、1 个滚动条、2 个标签、1 个计时器和 1 个图像框, 如图 2-4 所示。属性设置如表 2-3。




图 2-4 窗体设置界面

表 2-3 控件属性设置

对象类型	对象名称	属性名	属性值
命令按钮	Command1	Caption	开始
命令按钮	Command2	Caption	停止
标签	Label1	Caption	快
标签	Label2	Caption	慢

续表

对象类型	对象名称	属性名	属性值
滚动条	HScroll1	Max	1000
	HScroll1	Min	1
	HScroll1	LargeChange	200
	HScroll1	SmallChange	20
图像框	Image1	Stretch	true
	Image1	picture	添加一幅图片

(2) 程序代码如下：

```
Private Sub Command1_Click()  
    Timer1.Enabled = True           '使计时器有效  
    Timer1.Interval = HScroll1.Value '用滚动条的值设置计时器时间间隔  
    Command1.Enabled = False  
    Command2.Enabled = True  
End Sub  
  
Private Sub Command2_Click()  
    Timer1.Enabled = False          '使计时器无效  
    Command1.Enabled = True  
    Command2.Enabled = False  
End Sub  
  
Private Sub HScroll1_Change()  
    Timer1.Interval = HScroll1.Value  
End Sub  
  
Private Sub Timer1_Timer()  
    If Image1.Left + Image1.Width > 0 Then  
        Image1.Move Image1.Left - 100  
    Else  
        Image1.Left = Form1.ScaleWidth  
    End If  
End Sub
```

运行结果如图 2-5 所示。

4) 简单文本设置

(1) 界面设计。建立窗体，设计界面，如图 2-6 所示调整控件大小和位置，使界面美观大方。




图 2-5 运行界面



图 2-6 窗体对象属性设置

(2) 属性设置。窗体对象属性如表 2-4 所示, 结果如图 2-7 所示。

表 2-4 窗体对象属性

对象	属性	设置值
Form	Caption	设置文本样式和字号
TextBox	Text	欢迎使用 VB6.0
	Scrollbars	2-Vertical
	Multiline	True
Frame1	名称	Framestyle
	Caption	样式
Frame2	名称	Framefont
	Caption	字号
CheckBox1	Caption	粗体
CheckBox2	Caption	斜体
CheckBox3	Caption	下划线
OptionButton1	Caption	8
OptionButton2	Caption	12
OptionButton3	Caption	24

(3) 编写程序代码。程序代码如下:

```
Private Sub Form_Load()
    Text1.FontBold= False
    Text1.FontItalic= False
    Text1.FontUnderline= False
    Text1.FontSize= 12
    Option2.Value= True
End Sub
Private Sub Check1_Click()
    Text1.FontBold= Check1.Value
End Sub
Private Sub Check2_Click()
    Text1.FontItalic= Check2.Value
End Sub
Private Sub Check3_Click()
    Text1.FontUnderline= Check3.Value
End Sub
Private Sub Option1_Click()
    Text1.FontSize= 8
End Sub
Private Sub Option2_Click()
    Text1.FontSize= 12
End Sub
```




图 2-7 显示结果

```
Private Sub Option3_Click()  
    Text1.FontSize= 24  
End Sub
```

5) 左右移动字幕

(1) 界面设计。新建一个标准的 EXE 工程，添加一个窗体，并按如图 2-8 所示把标签、命令按钮、文本框、单选按钮、框架、计时器等控件添加到窗体上。




图 2-8 窗体对象属性设置

(2) 属性设置，如表 2-5 所示。

表 2-5 窗体对象属性

对象	属性	设置值
Form1	Caption	滚动字幕
Label2	Caption	移动速度
Label3	Caption	移动幅度
OptionButton1	Caption	Left
OptionButton2	Caption	Right
Command1	Caption	&S 暂停
Command2	Caption	&E 退出

(3) 编写程序代码。本程序的重点是如何控制字幕的移动，方法是每隔一定的时间来改变用来显示文本的标签左边的位置，从而达到移动字幕的效果。为控制字幕的移动，还需解决移动方向、移动速度和移动幅度等具体参数的位置。

程序代码如下：

```
Dim direction, value  
'direction 用于方向控制,1 向左移,2 向右移; value 为每次移动的距离  
Private Sub Form_Load()  
    Label1.Caption = "天长地久有时尽,此恨绵绵无绝期."  
    HScroll11.Min = 10: HScroll11.Max = 100  
    HScroll11.LargeChange = 10  
    Text1.Text = "20"
```

```
Option1.value = False
Option2.value = True
End Sub

Private Sub Command1_Click() '命令按钮 1 实现字幕移动与暂停的目的
    If Command1.Caption = "& S 暂停" Then
        Command1.Caption = "& C 继续"
        Timer1.Enabled = False '计时器失效,暂停移动
    Else
        Command1.Caption = "& S 暂停"
        Timer1.Enabled = True
    End If
End Sub

Private Sub HScroll1_Change()
    Timer1.Interval = 101 - HScroll1.value
End Sub

Private Sub Option1_Click()
    direction = 1
End Sub

Private Sub Option2_Click()
    direction = 2
End Sub

Private Sub Text1_Change()
    value = Val(Text1.Text)
End Sub

Private Sub Timer1_Timer()
    If direction = 1 Then
        If Label1.Left + Label1.Width > 0 Then
            Label1.Move Label1.Left - value
        Else
            Label1.Left = Form1.ScaleWidth
        End If
    Else
        If Label1.Left < Form1.ScaleWidth Then
            Label1.Move Label1.Left + value
        Else
            Label1.Left = 0 - Label1.Width
        End If
    End If
End Sub

Private Sub Command2_Click()
    Unload Me '或 End
End Sub
```

(4) 调试与运行。保存工程文件,然后运行程序,得到如图 2-9 所示的运行界面。